Maple commands define a function

```
> f:=x*(1-x-b*y)+y*(2-y-c*x);
```

$$f := x\,(1 - x - b\,y) + y\,(2 - y - c\,x) \tag{1}$$

Take the derivatives of this wrt x and y

```
> dx:=diff(f,x);
```

$$dx := 1 - 2\,x - b\,y - y\,c \tag{2}$$

```
> dy:=diff(f,y);
```

$$dy := -x\,b + 2 - 2\,y - c\,x \tag{3}$$

Solve for dx=dy=0 for (x,y) put multiple equations in {}

```
> ans:=simplify(solve({dx,dy},{x,y}));
```

$$ans := \left\{x = \frac{2\,(-1 + b + c)}{-4 + b^2 + 2\,b\,c + c^2},\ y = \frac{-4 + c + b}{-4 + b^2 + 2\,b\,c + c^2}\right\} \tag{4}$$

Note that simplify(stuff) and expand(stuff) can lead to more readable stuff. I will now copy the answers into new variables

```
> xs:=(2*(-1+b+c))/(-4+b^2+2*b*c+c^2);
```

$$xs := \frac{2\,(-1 + b + c)}{-4 + b^2 + 2\,b\,c + c^2} \tag{5}$$

```
> ys:=(-4+c+b)/(-4+b^2+2*b*c+c^2);
```

$$ys := \frac{-4 + c + b}{-4 + b^2 + 2\,b\,c + c^2} \tag{6}$$

I'll plot both of these as a function of c for b fixed using the subs command - note ys is the larger one:

```
> plot({subs(b=.01,xs),subs(b=.01,ys)},c=0..0.2);
```




```
> plot({subs(b=0.1,xs),subs(b=0.1,ys)},c=0..0.2);
```



Sensitivity of xs to b:

```
> sxb:=simplify((b/xs)*diff(xs,b));
```

$$sxb := -\frac{\left(4 + b^2 + 2\,b\,c + c^2 - 2\,b - 2\,c\right) b}{\left(-4 + b^2 + 2\,b\,c + c^2\right)\left(-1 + b + c\right)} \tag{7}$$

```
> sxb0:=subs(c=.25,b=.1,sxb);
```

$$sxb0 := -0.1357932847 \tag{8}$$

Do an integral:

```
> int(sin(2*x)*cos(x)^3,x);
```

$$-\frac{1}{40}\cos(5\,x) - \frac{1}{4}\cos(x) - \frac{1}{8}\cos(3\,x) \tag{9}$$

```
> int(sin(2*x)*cos(x)^3,x=0..Pi/2);
```

$$\frac{2}{5} \tag{10}$$

solve a differential equation:

```
> dsolve({diff(x(t),t)=x(t)*(1-x(t)/5),x(0)=1},x(t));
```

$$x(t) = \frac{5}{1 + 4\,e^{-t}} \tag{11}$$